>

Volume of a Solid

```
> restart:with(plots):
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,
  70]);
```

Page 1055 # 18

We wish to find the volume of the solid bounded by $z = 2x + y + 1$ (red), $z = -2x$ (green), $x = y^2$ (blue), and $x = 1$ (gold). We find the intersection of the first two planes.

```
> z1:=2*x+y+1;
  z2:=-2*x;
  yline1:=solve(z1=z2,y);
```

$$z1 := 2x + y + 1$$

$$z2 := -2x$$

$$yline1 := -4x - 1$$

We plot this line along with the intersections of the cylinder $x = y^2$ and the plane $x = 1$ with the xy-plane.

```
> p1:=plot(yline1,x=-1..1.5,y=-1.5..1.5):
  p2:=implicitplot(x=y^2,x=-1..1.5,y=-1.5..1.5):
  p3:=implicitplot(x=1,x=-1..1.5,y=-1.5..1.5):
  display(p1,p2,p3);
```



Since the line of intersection of the two planes does not intersect the enclosed region on the right, one plane will always be the higher over that region. We can evaluate each plane at (0,0) to see which that will be.

```
> z1value:=eval(z1,{x=0,y=0});
  z2value:=eval(z2,{x=0,y=0});
```

$$z1value := 1$$

$$z2value := 0$$

Thus $z = 2x + y + 1$ is the higher of the two planes over our region. We look at the graphs and the resulting solid.

```
> p1:=implicitplot3d(z=2*x+y+1,x=-1..1,y=-1..1,z=-5..5,style=
  patchnogrid,color=red):
  p2:=implicitplot3d(z=-2*x,x=-1..1,y=-1..1,z=-5..5,style=
  patchnogrid,color=green):
  p3:=implicitplot3d(x=y^2,x=-1..1,y=-1..1,z=-5..5,style=
  patchnogrid,color=blue):
  p4:=implicitplot3d(x=1,x=-1..1.1,y=-1..1,z=-5..5,style=
  patchnogrid,color=gold):
  display(p1,p2,p3,p4);
```



To get a better look at the interior of the solid, we remove the gold plane.

```
> display(p1,p2,p3);
```



We are looking for the volume between the two planes.

```
> V:=int(int((2*x+y+1)-(-2*x),x=y^2..1),y=-1..1);
```

$$V := \frac{68}{15}$$

>